Global Trade Analysis Project

Overview of GTAP-RD (Recursive Dynamic) Model

Angel Aguiar, **Erwin Corong** and Dominique van der Mensbrugghe
Center for Global Trade Analysis

2020 GTAP Conference

Outline

- Motivation
- Static and Dynamic model characteristics
- Overview of year-on-year (dynamic) capital accumulation
- Additional core (static) model features
- Dynamic features

Motivation

- **Dynamic specification:**
  - Can easily be added to the standard GTAP (classic or GTAPv7) model
  - Focus on baseline creation and trade projections
  - Additional dynamic features (e.g., cost-neutral twist for production and Armington demands à la Dixon and Rimmer(2002))
  - With levers to accommodate inputs from other (e.g., Macro) models

- **Starting point: Standard GTAP (GTAPv7) model**

Static Model

- Move from one equilibrium point to another
- Results are mostly due to efficiency effects—i.e., reallocation of resources
- Fixed aggregate endowments (unless shocked or slack factor market (i.e., unemployment) closure)
- Investment changes only affect the composition of demand (no impact on beginning period capital stock)

Recursive Dynamic (RD) Model

- **Sequence of time-based comparative static equilibria (with myopic expectations)**
- **Endogenous capital accumulation to drive growth in capital stock**
  - Stock-flow relation between investment and capital stock, typically with a (1-year) gestation lag
- **Between period updates to:**
  - Exogenous endowments (population, labor force)
  - Technology
  - Preferences
  - Policy

GTAP-RD model development

- **GTAP version 7 (facelift) model with additional features**
- **Dynamic equations added at the bottom of core GTAP code**
  - A few slack or technology variables added in the core GTAP code
- **Maintain model readability**
  - Consistent variable, coefficient and equation naming convention
- **Replicability: GEMPACK and GAMS version**
  - Fisher true price indices
- **World Trade Organization (WTO)**
  - WTO Global Trade Model (Aguiar, Corong, van der Mensbrugghe, Bekkers, Teh and Koopman, 2019)

New GTAP (core) Model Features

- **Additional flexibility**
  - Price indices (MUV, domestic absorption by region)
  - Capital account closure
  - Labor-biased technical change
  - Region-wide technical change variables
    - Labor only
    - Non-capital factor productivity (from the Center's GDyn model)
  - Upward sloping factor supply curves
    - Labor and land
    - Sector-specific (natural resource) factor

Capital accumulation

- **Capital accumulation**

$$K_{t+1} = (1-\delta)K_t + I_t$$

$$\Delta K_t = I_t - \delta K_t$$

- **In the standard GTAP Model code:**

$$KE = KB*(1-\delta) + QINV$$

$$ke = (1 - INVKERATIO)*kb + INVKERATIO*qinv$$

Static GTAP Model

- Capital accumulation function determines end of period capital stock

$$ke = (1 - INVKERATIO) * kb + INVKERATIO * qinv$$

- In Static GTAP, the beginning period capital stock (kb) does not change because aggregate capital supply, qe("capital",r) is fixed

```
Equation E_kb
# associates change in cap. services w/ change in cap. stock #
(all,r,REG)
    kb(r) = sum{e,ENDWC, [VES(e,r) / GROSSCAP(r)] * qe(e,r)};
```

Capital Accumulation

- **Add an equation that will link net investment changes within the year to year-on-year growth in**
  - Aggregate capital endowment, **qe(“capital”,r)**
  - Beginning period capital stock, **kb**

Change in Capital Stock (Preliminaries: Variables)

```
Variable (change)
    del_unity # shifter for activating recursive dynamics #;
Variable (all,r,REG)
    capadd(r) # net capital addition (growth) within period #;
Variable (all,r,REG)
    qecap(r) # aggregate capital endowment in reqion r #;
```

Change in Capital Stock (Preliminaries: Coefficients)

```
• Coefficient (all,r,REG)
      DEPR(r) # Depreciation rate #;
  Coefficient (all,r,REG)
      VKBINI(r) # beginning-of-period capital stock (initial value) #;
  Coefficient (parameter)(all,r,REG)
      VKADD(r) # addition to capital stock from previous period's
  investment #;

• Formula (all,r,REG)
      DEPR(r) = VDEP(r) / VKB(r);
  Formula (initial)(all,r,REG)
      VKBINI(r) = VKB(r);
  Update (all,r,REG)
      VKBINI(r) = qecap(r);
  Formula (initial)(all,r,REG)
      VKADD(r) = REGINV(r) - [VKB(r) * DEPR(r)];
```

Change in beginning period Capital Stock

- In the core GTAP model code, VKB and kb are formulated as:

```
Update (all,r,REG)
    VKB(r) = pinv(r) * kb(r);

Equation E_kb
# associates change in cap. services w/ change in cap. stock #
(all,r,REG)
    kb(r) = sum{e,ENDWC, [VES(e,r) / GROSSCAP(r)] * qe(e,r)};
```

- So In GTAP-RD, we add:

```
Equation E_capadd
# capital addition equation #
 (all,r,REG)
    0.01 * VKBINI(r) * qecap(r) = VKADD(r) * del_unity + capadd(r);


Equation E_qecap
# aggregate capital endowment in region r #
(all,r,REG)
    qecap(r) = sum{e,ENDWC, [VES(e,r) / sum{c,ENDWC, VES(c,r)}] * qe(e,r)};
```

Change in beginning period Capital Stock (2)

- In the core GTAP model code, VKB and kb are formulated as:

```
Update (all,r,REG)
    VKB(r) = pinv(r) * kb(r);

Equation E_kb
# associates change in cap. services w/ change in cap. stock #
(all,r,REG)
    kb(r) = qe("capital",r);
```

- So In GTAP-RD, we add:

```
Equation E_capadd
# capital addition equation #
 (all,r,REG)
     qecap(r) = 100 * [VKADD(r) / VKBINI(r)] * del_unity (1.0) + capadd(r);


Equation E_qecap
# aggregate capital endowment in region r #
(all,r,REG)
    qecap(r) = qe("capital",r);
```

CES Twists

- **Production function**
  - Change the capital/labor ratio by ‘tw’ percent, assuming cost neutrality
- **Armington demands**
  - Change the import/domestic ratio by ‘tw’ percent, assuming cost neutrality
    - Firms
    - Private households
    - Investment demand
    - government demand
  - Second-level: change the ratio of a group of source regions relative to another group of source regions, assuming cost neutrality

Intuition behind ‘twist’ Specification



- **Isoquant AA**
  - X is the chosen or optimal combination of domestic and imported goods
- **Preference changes towards imports translate both AA down and to right, such that BB still passes through X—i.e., AA has been twisted around X.**

CES Twist: Household (dom-imp) demands

```
Equation E_qpd
# private consumption demand for domestic goods #
(all,c,COMM)(all,r,REG)
    qpd(c,r) = qpa(c,r) - ESUBD(c,r) * [ppd(c,r) - ppa(c,r)] + apd(c,r);

Equation E_qpm
# private consumption demand for aggregate imports #
(all,c,COMM)(all,r,REG)
    qpm(c,r) = qpa(c,r) - ESUBD(c,r) * [ppm(c,r) - ppa(c,r)] + apm(c,r);


Equation E_apd
# equation to facilitate shift towards cons of dom. c by priv hhld in region r #
(all,c,COMM)(all,r,REG)
    apd(c,r) = PMSHR(c,r) * apdmtwist(c,r);

Equation E_apm
# equation to facilitate shift towards cons of imp. c by priv hhld in region r #
(all,c,COMM)(all,r,REG)
    apm(c,r) = [PMSHR(c,r) - 1] * apdmtwist(c,r);

Equation E_apdmtwist
# equation containing all twist towards cons of dom. c by priv hhld in reg r#
(all,c,COMM)(all,r,REG)
    apdmtwist(c,r) = apdmtwistall(c,r) + apdmtwistcom(c) + apdmtwistreg(r) + admtwist(c,r);
```

CES Twist: Firms' (dom-imp) demand

```
Equation E_qfd
# act. a demands for domestic good c #
(all,c,COMM)(all,a,ACTS)(all,r,REG)
    qfd(c,a,r) = qfa(c,a,r) - ESUBD(c,r) * [pfd(c,a,r) - pfa(c,a,r)] + afd(c,a,r);

Equation E_qfm
# act. a demands for composite import c #
(all,c,COMM)(all,a,ACTS)(all,r,REG)
    qfm(c,a,r) = qfa(c,a,r) - ESUBD(c,r) * [pfm(c,a,r) - pfa(c,a,r)] + afm(c,a,r);

Equation E_afd
# equation to facilitate shift towards domestic input c in act. a in region r #
(all,c,COMM)(all,a,ACTS)(all,r,REG)
    afd(c,a,r) = FMSHR(c,a,r) * afdmtwist(c,a,r);

Equation E_afm
# equation to facilitate shift towards imported input c in act. a in region r #
(all,c,COMM)(all,a,ACTS)(all,r,REG)
    afm(c,a,r) = [FMSHR(c,a,r) - 1] * afdmtwist(c,a,r);

Equation E_afdmtwist
# shift towards domestic interm. use by act a in reg r #
(all,c,COMM)(all,a,ACTS)(all,r,REG)
    afdmtwist(c,a,r) = afdmtwistall(c,a,r) + afdmtwistact(a) + afdmtwistcom(c) + afdmtwistreg(r) +
admtwist(c,r);
```

CES Twist: Labor-Capital

```
Equation E_qfe
# demands for endowment commodities #
(all,e,ENDW)(all,a,ACTS)(all,r,REG)
    qfe(e,a,r)
        = IF[e in ENDWLC,
             - afe(e,a,r) + [afetwist(e,a,r) - afetwistave(a,r)] + qva(a,r)
             - ESUBVA(a,r) * [pfe(e,a,r) - afe(e,a,r) - pva(a,r)]]
       + IF[e in ENDWSF,
             - afe(e,a,r) + qva(a,r)
             - ESUBVA(a,r) * [pfe(e,a,r) - afe(e,a,r) - pva(a,r)]];

Equation E_pva
# effective price of primary factor composite in each sector/region #
(all,a,ACTS)(all,r,REG)
    pva(a,r) = sum{e,ENDW, VASHR(e,a,r) * [pfe(e,a,r) - afe(e,a,r)]};

Equation E_afetwistave
# average endowment twist shifter in activity a in reg r: to ensure cost neutrality #
(all,a,ACTS)(all,r,REG)
    0 = sum{e,ENDWLC, VFP(e,a,r) * [afetwist(e,a,r)  - afetwistave(a,r)]};

Equation E_afetwist
# sector and region specific twist towards use of endowment e #
(all,e,ENDWLC)(all,a,ACTS)(all,r,REG)
    afetwist(e,a,r) = afetwistall(e,a,r) + afetwistact(e,a) + afetwistreg(e,r);
```

CES Twist: across trading partners

```
Equation E_qxs
# regional demand for disaggregated imported commodities by source #
(all,c,COMM)(all,s,REG)(all,d,REG)
    qxs(c,s,d)
        = -ams(c,s,d) + [amdstwist(c,s,d) - amdstwistave(c,d)]
        + qms(c,d) - ESUBM(c,d) * [pmds(c,s,d) - ams(c,s,d) - pms(c,d)];

Equation E_amdstwistave
# average top-level import twist shifter for com c by importing reg d #
(all,c,COMM)(all,d,REG)
    0 = sum{s,REG, VMSB(c,s,d) * [amdstwist(c,s,d) - amdstwistave(c,d)]};

Equation E_amdstwist
# commodity and region specific twist towards imports of com c from source s
#
(all,c,COMM)(all,s,REG)(all,d,REG)
    amdstwist(c,s,d) = amdstwistall(c,s,d) + amdstwistsrc(c,s) +
amdstwistreg(s,d);
```

Additional region-specific tech change

```
!< GTAPv7-RD: input tech change and shifters for all factors except capital >!
Variable (all,e,ENDWXC)(all,a,ACTS)(all,r,REG)
    afendwxc(e,a,r) # non-capital factor e augmenting tech change by a in r #;
Variable (all,a,ACTS)(all,r,REG)
    afendwxcact(a,r) # non-capital factor augmenting tech change by a in r #;
Variable (all,r,REG)
    afendwxcreg(r) # non-capital factor tech change in region r #;

Equation E_afe
# sector/region specific average rate of prim. factor e augmenting tech change #
(all,e,ENDW)(all,a,ACTS)(all,r,REG)
    afe(e,a,r)
        = afecom(e) + afesec(a) + afereg(r) + afeall(e,a,r) + afecomreg(e,r)
!< GTAPv7-RD: input technical change and shifters >!
        + IF[e in ENDWL,  afelabreg(r) + afelabact(a,r) + afelab(e,a,r)]
        + IF[e in ENDWXC, afendwxcreg(r) + afendwxcact(a,r) + afendwxc(e,a,r)];
```

Labor-biased technical change

- Many potential labor productivity factors as there are activities and labor types

$$\pi_{l,a}^{l} = \alpha_{l,a}^{l} + \beta_{l,a}^{l}\gamma^{l}$$

Where:

γ^{l} economy-wide parameter that can be used to target GDP

$\alpha_{l,a}^{l}$ additive labor productivity

$\beta_{l,a}^{l}$ multiplicative labor productivity

Uniformity: $\alpha_{l,a}^{l}$=0 and $\beta_{l,a}^{l}$=1 for all skill types

Labor-biased tech change in TABLO

```
Variable (all,e,ENDWL)(all,a,ACTS)(all,r,REG)
    afelabadd(e,a,r) # labor productivity additive shifter #;
Coefficient (parameter)(all,e,ENDWL)(all,a,ACTS)(all,r,REG)
    LABMULT(e,a,r) # labor productivity multiplicative parameter #;

Formula (initial) (all,e,ENDWL)(all,a,ACTS)(all,r,REG)
    LABMULT(e,a,r) = 1.0; ! ie, Default 1.0 !
Read (ifheaderexists)
    LABMULT from file GTAPPARM header "LMLT";
Assertion (all,e,ENDWL)(all,a,ACTS)(all,r,REG)
    LABMULT(e,a,r) ge 1.0;


Equation E_afelab
# labor productivity shifters #
(all,e,ENDWL)(all,a,ACTS)(all,r,REG)
      afelab(e,a,r) = afelabadd(e,a,r) + [LABMULT(e,a,r) *
afelabreg(r)];
```

Factor market equilibrium

- **(Perfectly) mobile endowments** (typically labor and capital)
  - Aggregate demand is equal to (exogenous) aggregate supply
- **Sluggish endowments** (typically land)
  - (Exogenous) aggregate supply is allocated across activities with a CET specification (ETRAE)
  - Activity supply is equal to activity demand
- **Sector-specific resources** (typically natural resources)
  - (Exogenous) activity supply is equal to activity demand
- **GTAP-RD relaxes assumption of exogenous supply for labor, land and sector-specific resources**

Closure

- **Investment is savings-driven**

$$PINV_r QINV_r = SAVE_r + FSAVE_r - DEPR_r$$

- **Four possible closures for foreign savings**
  - Foreign savings adjust to changes in regional rates of return, assumes equalization of expected rate of return
  - Global investment maintains same allocation share as in base year
  - Foreign savings are fixed in real terms
  - Foreign savings as a share of GDP are fixed (requires a residual region)

$$\underbrace{\sum_c \sum_d PFOB_{c,r,d} QXS_{c,r,d}}_{\text{Exports}} + \underbrace{VST_r}_{\text{Margin exports}} + \underbrace{FSAVE_r}_{\text{Foreign savings}} = \underbrace{\sum_c \sum_s PCIF_{c,s,r} QXS_{c,s,r}}_{\text{Imports}}$$

Prices

- **Moved to chained Fisher ‘ideal’ price indices**
  - Geometric average of Laspeyres and Paasche price indices

$$F_t^P = F_{t-1}^P \sqrt{\underbrace{\frac{\sum_i P_{i,t} Q_{i,t-1}}{\sum_i P_{i,t-1} Q_{i,t-1}}}_{\text{Lagged volume weights}} \underbrace{\frac{\sum_i P_{i,t} Q_{i,t}}{\sum_i P_{i,t-1} Q_{i,t}}}_{\text{Current volume weights}}} = F_{t-1}^P \sqrt{L_t^P P_t^P}$$

References

Aguiar, A., E. Corong, and D. van der Mensbrugghe (2019). “The GTAP Recursive Dynamic (GTAP-RD) Model: Version 1.0”. Center for Global Trade Analysis, Purdue University. https://mygeohub.org/groups/gtap/dynamic-docs

Aguiar, A., E. Corong, D. van der Mensbrugghe, E. Bekkers, R. Koopman (2019) and R. Teh. “The WTO Global Trade Model” Technical documentation”. Staff Working Paper ERSD-2019-10. The World Trade Organization, Geneva.

https://www.wto.org/english/res_e/reser_e/ersd201910_e.htm

Dixon, P.B., and M.T. Rimmer (2002). “Dynamic general equilibrium modelling for forecasting and policy: a practical guide and documentation of MONASH”, 1st ed., vol. 256. Amsterdam: Elsevier.

Aside: Hertel and Tsigas chapter 2

$$KE = KB*(1-DEPR)+QINV$$

$$dKE = (1-DEPR)\,dKB + dQINV$$

$$ke = (1-DEPR)*\left(\frac{KB}{KE}\right)*kb + \left(\frac{QINV}{KE}\right)*qinv$$

$$INVKERATIO = PINV*\left(\frac{QINV}{KE}\right) = REGINV\,/\,VKE$$

$$(1-DEPR)*\left(\frac{KB}{KE}\right) = \left[VKB*(1-DEPR)+REGINV-REVGINV\right]/VKE$$

$$= \left[VKE-REVGINV\right]/VKE$$

$$= 1-INVKERATIO$$

$$ke = (1-INVKERATIO)*kb + INVKERATIO*qinv$$

Appendix: Dynamic capital accumulation

- **Standard motion equation**

$$K_{t+1} = (1-\delta)^n K_t + \frac{(1+g)^n - (1-\delta)^n}{g+\delta} I_t$$

Where:

$$g = \left(\frac{I_{t+1}}{I_t} \right)^{1/n} - 1$$

n = number of periods

Appendix: Derivation of (between period) change in beginning capital stock

- **Change in capital stock in period t**

$$\Delta K_t = QINV_t - \delta * KB_t$$

- **Multiply by price of capital, pinv**

$$\Delta K_t * pinv = QINV_t * pinv - \delta * KB_t * pinv$$

- **In percentage change form, where k is the percentage change in K (so kK/100=ΔK):**

$$0.01*[K * pinv_t]*k = \left(QINV_t * pinv - \delta * KB_t * pinv\right)*del\_Unity$$

Where: del_Unity is exogenous

- **This equation means that the change in capital stock is equal to net investment**

Global Trade Analysis Project

Overview of GTAP-RD (Recursive Dynamic) Model

Comments and Questions: ecorong@purdue.edu

Center for Global Trade Analysis
Department of Agricultural Economics, Purdue University
403 West State Street, West Lafayette, IN 47907-2056 USA